

【教材 5.1】试编写一程序，将以变量 BUF 为首址的 100 个字节存储单元清零。

【解答】使用串操作指令实现这一功能。程序如下：

```
.386
DATA SEGMENT USE16
BUF DB a1,a2,...,a100
DATA ENDS
STACK SEGMENT USE16 STACK
      DB 200H DUP(0)
STACK ENDS
CODE SEGMENT USE16
      ASSUME CS:CODE, SS:STACK, ES: DATA
START: MOV AX,DATA
      MOV ES, AX
      LEA DI, BUF
      MOV AL,0
      MOV CX,100; 前述语句均为为使用串操作指令做准备
      REP STOSB ;将 0 重复送入以 BUF 为首址的 100 个字节中
      MOV AX,4C00H
      INT 21H
CODE ENDS
      END START
```

【教材 5.2】设有一个数组 a_i ($i=1, 2, \dots, 1000$) 存放在 A 为起始地址的字存储区中，现要求将数组加以压缩，使其中的非 0 元素仍按序存放在 A 存储区中，而 0 元素不再出现，试用串操作指令编写实现上述功能的程序。

【解答】

解题思路：若数组存放在以 ARR 为起始地址的存储区中，采用 SI,DI 两个数据指针，一个取数，一个存数。逐一将 ARR 存储区中的数据取出，判断是否为 0，若非 0 则存回去，否则就将该数过滤掉，继续取下一个数处理，直到所有的数据处理完毕。

寄存器的分配：

SI：取数指针；

DI：存数指针；

AX：存放临时取出的数据

CX:数据的个数。

程序如下：

```
.386
DATA SEGMENT USE16
ARR DW a1, a2, 0, a3, 0,..., 0, a1000
DATA ENDS
STACK SEGMENT USE16 STACK
      DB 200H DUP(0)
STACK ENDS
CODE SEGMENT USE16
```

```

        ASSUME CS:CODE,  SS:STACK, DS:DATA,  ES:DATA
START: MOV     AX,      DATA
        MOV     DS,     AX
        MOV     ES,     AX
        LEA     SI,     ARR
        LEA     DI,     ARR
        MOV     CX,     1000        ; 准备使用串操作指令
NEXT:  LODSW                                ; 取一数  AX
        CMP     AX,     0
        JE      NOSTO        ; 如果该数为 0，则不送入原数组中，转
NOSTO
        STOSW                                ;该数不为 0，则送回原数组中
NOSTO: LOOP NEXT
        MOV     AX,4C00H
        INT     21H
CODE  ENDS
        END START

```

【教材 5.6】① 编写一宏定义 IO，要求实现 9 号与 10 号系统功能调用的功能。

② 请将以下语句按宏定义 IO 的要求用宏调用语句代替。

```

        LEA     DX,     BUF1
        MOV     AH,     9
        INT     21H

        .....

        LEA     DX,     BUF2
        MOV     AH,     10
        INT     21H

```

【解答】

(1)宏定义 IO 定义如下：

```

IO MACRO  A, B
    LEA  DX,  A
    MOV  AH,  B
    INT  21H
ENDM

```

(2)宏调用如下：

```

IO  BUF1, 9
.....
IO  BUF2, 10

```

【教材 5.9】已知宏定义如下：

.386

```

OUT_CHR MACRO  A
    MOV  DL,  A
    MOV  AH,  2

```

```

                INT    21H
                ENDM
OUT_STR MACRO   A, B
                LOCAL   C
                MOV     CX, B
C:    OUT_CHR A
                LOOP    C
            ENDM

```

调用以上宏定义实现以下功能：从键盘输入一个数(1-9)，紧跟其后显示相同个数的*后返回 DOS；若输入的不是 1-9，则给出错误提示后重输，直到输入正确为止。

【解答】

解题思路：给出提示，输入一个数，采用 1 号 DOS 系统功能调用输入一个字符，判断该字符是否在 1-9 之间，若在则将该字符转换为数值 A，输出 A 个*后结束，否则再次提示输入，直到正确为止。

程序如下：

```

.386
OUT_CHR MACRO   A
                MOV     DL,    A
                MOV     AH,    2
                INT     21H
                ENDM
OUT_STR MACRO   A, B
                LOCAL   C
                MOV     CX, B
C:    OUT_CHR A
                LOOP    C
            ENDM

STACK SEGMENT USE16  STACK
                DB 200 DUP(0)
STACK ENDS
DATA SEGMENT USE16
BUF1  DB 0AH, 0DH, 'PLEASE INPUT A NUMBER', 0AH, 0DH, '$'
ERROR DB 0AH, 0DH, 'ERROR!', 0AH, 0DH, '$'
DATA ENDS
CODE SEGMENT USE16
        ASSUME     CS:CODE, SS:STACK, DS:DATA
BEGIN: MOV     AX,    DATA
        MOV     DS,    AX
L1:     LEA     DX,    BUF1
        MOV     AH,    9
        INT     21H
        MOV     AH,    1

```

```
        INT     21H
        CMP     AL, '1'
        JB      ERR
        CMP     AL, '9'
        JA      ERR
        SUB     AL, 30H
        MOVZX   AX, AL
        OUT_STR '*', AX
        MOV     AH, 4CH
        INT     21H
ERR:    LEA     DX, ERROR
        MOV     AH, 9
        INT     21H
        JMP     L1
CODE ENDS
END BEGIN
```